Problem:
Score : 400 points

Problem Statement

There is a grid with H horizontal rows and W vertical columns, and there are obstacles on some of the squares.

Snuke is going to choose one of the squares not occupied by an obstacle and place a lamp on it.
The lamp placed on the square will emit straight beams of light in four cardinal directions: up, down, left, and right.
In each direction, the beam will continue traveling until it hits a square occupied by an obstacle or it hits the border of the grid. It will light all the squares on the way, including the square on which the lamp is placed, but not the square occupied by an obstacle.

Snuke wants to maximize the number of squares lighted by the lamp.

You are given H strings S_i (1 \leq i \leq H), each of length W. If the j-th character (1 \leq j \leq W) of S_i is #, there is an obstacle on the square at the i-th row from the top and the j-th column from the left; if that character is ., there is no obstacle on that square.

Find the maximum possible number of squares lighted by the lamp.

Constraints

1 \leq H \leq 2,000

1 \leq W \leq 2,000

S_i is a string of length W consisting of # and ..

. occurs at least once in one of the strings S_i (1 \leq i \leq H).

Input

Input is given from Standard Input in the following format:

H W
S_1
:
S_H

Output

Print the maximum possible number of squares lighted by the lamp.

Sample Input 1

4 6
#..#..
.....#
....#.
#.#...

Sample Output 1

8

If Snuke places the lamp on the square at the second row from the top and the second column from the left, it will light the following squares: the first through fifth squares from the left in the second row, and the first through fourth squares from the top in the second column, for a total of eight squares.

Sample Input 2

8 8
..#...#.
....#...
##......
..###..#
...#..#.
##....#.
#...#...
###.#..#

Sample Output 2

13

Solution in Haskell (GHC 8.8.3) (judge verdict: Accepted)

import qualified Data.Text.Lazy as T
import qualified Data.Text.Lazy.IO as T
import qualified Data.Text.Lazy.Read as T
import qualified Data.Vector.Unboxed as VU
import qualified Data.Vector as V

main :: IO ()
main = do
  [h, w] <- map unsafeTextToInt . T.words <$> T.getLine :: IO [Int]
  let
    encode :: (Int, Int) -> Int
    encode (i, j) = i * w + j
    decode :: Int -> (Int, Int)
    decode ij = ij `divMod` w
  xs <- VU.fromList . map ('.' ==) . concat . lines <$> getContents :: IO (VU.Vector Bool)
  let
    ls = VU.constructN (h * w) rec
      where
        rec :: VU.Vector Int -> Int
        rec v | not (xs VU.! ij) = 0
              | 0 < j            = VU.last v + 1
              | otherwise        = 1
          where
            l = VU.length v :: Int
            ij = l :: Int
            (_, j) = decode ij
    rs = VU.constructrN (h * w) rec
      where
        rec :: VU.Vector Int -> Int
        rec v | not (xs VU.! ij) = 0
              | j < w - 1        = VU.head v + 1
              | otherwise        = 1
          where
            l = VU.length v :: Int
            ij = h * w - l - 1 :: Int
            (_, j) = decode ij
    us = VU.constructN (h * w) rec
      where
        rec :: VU.Vector Int -> Int
        rec v | not (xs VU.! ij) = 0
              | w <= l           = v VU.! (l - w) + 1
              | otherwise        = 1
          where
            l = VU.length v :: Int
            ij = l :: Int
    ds = VU.constructrN (h * w) rec
      where
        rec :: VU.Vector Int -> Int
        rec v | not (xs VU.! ij) = 0
              | w <= l           = v VU.! (w - 1) + 1
              | otherwise        = 1
          where
            l = VU.length v :: Int
            ij = h * w - l - 1 :: Int
    ys = op <$> [0 .. (h - 1)] <*> [0 .. (w - 1)]
      where
        op :: Int -> Int -> Int
        op i j = max 0 . subtract 3 $ ls VU.! ij + rs VU.! ij + us VU.! ij + ds VU.! ij
          where
            ij = encode (i, j)
  print . maximum $ ys

unsafeTextToInt :: T.Text -> Int
unsafeTextToInt s = case T.signed T.decimal s of
  Right (n, _) -> n
{-# INLINE unsafeTextToInt #-}